In [8]:
```julia
import Pkg
Pkg.activate(@__DIR__)
Pkg.instantiate()
import MathOptInterface as MOI
import Ipopt
import FiniteDiff
import ForwardDiff as FD
import Convex as cvx
import ECOS
using LinearAlgebra
using Plots
using Random
using JLD2
using Test
import MeshCat as mc
```

```
 Activating project at `~/OCRL/HW3_S25`
```

In [9]:
```julia
include(joinpath(@__DIR__, "utils","fmincon.jl"))
include(joinpath(@__DIR__, "utils","cartpole_animation.jl"))
```

```
animate_cartpole (generic function with 1 method)
```

NOTE: This question will have long outputs for each cell, remember you can use `cell -> all output -> toggle scrolling` to better see it all

Q1: Direct Collocation (DIRCOL) for a Cart Pole (30 pts)

We are now going to start working with the NonLinear Program (NLP) Solver IPOPT to solve some trajectory optimization problems. First we will demonstrate how this works for simple optimization problems (not trajectory optimization). The interface that we have setup for IPOPT is the following:

$$
\begin{aligned}
\min_{x} \quad & \ell(x) && \text{cost function} && (1)\\
\text{st} \quad & c_{eq}(x) = 0 && \text{equality constraint} && (2)\\
& c_L \leq c_{ineq}(x) \leq c_U && \text{inequality constraint} && (3)\\
& x_L \leq x \leq x_U && \text{primal bound constraint} && (4)
\end{aligned}
$$

where $\ell(x)$ is our objective function, $c_{eq}(x) = 0$ is our equality constraint, $c_L \leq c_{ineq}(x) \leq c_U$ is our bound inequality constraint, and $x_L \leq x \leq x_U$ is a bound constraint on our primal variable x.

Part A: Solve an LP with IPOPT (5 pts)

To demonstrate this, we are going to ask you to solve a simple Linear Program (LP):

$$
\begin{aligned}
\min_{x} \quad & q^T x && (5)\\
\text{st} \quad & Ax = b && (6)\\
& Gx \leq h && (7)
\end{aligned}
$$

Your job will be to transform this problem into the form shown above and solve it with IPOPT. To help you interface with IPOPT, we have created a function `fmincon` for you. Below is the docstring for this function that details all of the inputs.

In [10]:
```julia
"""
x = fmincon(cost,equality_constraint,inequality_constraint,x_l,x_u,c_l,c_u,x0,params,diff_type)

This function uses IPOPT to minimize an objective function

`cost(params, x)`

With the following three constraints:

`equality_constraint(params, x) = 0`
`c_l <= inequality_constraint(params, x) <= c_u`
`x_l <= x <= x_u`

Note that the constraint functions should return vectors.

```

```
Problem specific parameters should be loaded into params::NamedTuple (things like
cost weights, dynamics parameters, etc.).

args:
    cost::Function                    - objective function to be minimzed (returns scalar)
    equality_constraint::Function     - c_eq(params, x) == 0
    inequality_constraint::Function   - c_l <= c_ineq(params, x) <= c_u
    x_l::Vector                       - x_l <= x <= x_u
    x_u::Vector                       - x_l <= x <= x_u
    c_l::Vector                       - c_l <= c_ineq(params, x) <= x_u
    c_u::Vector                       - c_l <= c_ineq(params, x) <= x_u
    x0::Vector                        - initial guess
    params::NamedTuple                - problem parameters for use in costs/constraints
    diff_type::Symbol                 - :auto for ForwardDiff, :finite for FiniteDiff
    verbose::Bool                     - true for IPOPT output, false for nothing

optional args:
    tol                               - optimality tolerance
    c_tol                             - constraint violation tolerance
    max_iters                         - max iterations
    verbose                           - verbosity of IPOPT

outputs:
    x::Vector                         - solution

You should try and use :auto for your `diff_type` first, and only use :finite if you
absolutely cannot get ForwardDiff to work.

This function will run a few basic checks before sending the problem off to IPOPT to
solve. The outputs of these checks will be reported as the following:

---------checking dimensions of everything----------
---------all dimensions good------------------------
---------diff type set to :auto (ForwardDiff.jl)----
---------testing objective gradient-----------------
---------testing constraint Jacobian----------------
---------successfully compiled both derivatives-----
---------IPOPT beginning solve----------------------

If you're getting stuck during the testing of one of the derivatives, try switching
to FiniteDiff.jl by setting diff_type = :finite.
""";
```

In [11]:

```
@testset "solve LP with IPOPT" begin

    LP = jldopen(joinpath(@__DIR__,"utils","random_LP.jld2"))

    params = (q = LP["q"], A = LP["A"], b = LP["b"], G = LP["G"], h = LP["h"])

    # return a scalar
    function cost(params, x)::Real
        # TODO: create cost function with params and x
        cost = params.q' * x
        return cost
    end

    # return a vector
    function equality_constraint(params, x)::Vector
        # TODO: create equality constraint function with params and x
        eq = params.A * x - params.b
        return eq
    end

    # return a vector
    function inequality_constraint(params, x)::Vector
        # TODO: create inequality constraint function with params and x
        ineq = params.G * x - params.h
        return ineq
    end

    # TODO: primal bounds
    # you may use Inf, like Inf*ones(10) for a vector of positive infinity
    x_l = -Inf * ones(20)
    x_u = Inf * ones(20)
```

```
    # TODO: inequality constraint bounds
    c_l = -Inf * ones(20)
    c_u = 0 * ones(20)

    # initial guess
    x0= randn(20)

    diff_type = :auto   # use ForwardDiff.jl
#     diff_type = :finite # use FiniteDiff.jl

    x = fmincon(cost, equality_constraint, inequality_constraint,
                x_l, x_u, c_l, c_u, x0, params, diff_type;
                tol = 1e-6, c_tol = 1e-6, max_iters = 10_000, verbose = true);


    @test isapprox(x, [-0.44289, 0, 0, 0.19214, 0, 0, -0.109095,
                       -0.43221, 0, 0, 0.44289, 0, 0, 0.192142,
                       0, 0, 0.10909, 0.432219, 0, 0], atol = 1e-3)

end
```

```
---------checking dimensions of everything----------
---------all dimensions good------------------------
---------diff type set to :auto (ForwardDiff.jl)----
---------testing objective gradient-----------------
---------testing constraint Jacobian----------------
---------successfully compiled both derivatives-----
---------IPOPT beginning solve----------------------
This is Ipopt version 3.14.17, running with linear solver MUMPS 5.7.3.

Number of nonzeros in equality constraint Jacobian...:       80
Number of nonzeros in inequality constraint Jacobian.:      400
Number of nonzeros in Lagrangian Hessian.............:        0

Total number of variables............................:       20
                     variables with only lower bounds:        0
                variables with lower and upper bounds:        0
                     variables with only upper bounds:        0
Total number of equality constraints.................:        4
Total number of inequality constraints...............:       20
        inequality constraints with only lower bounds:        0
   inequality constraints with lower and upper bounds:        0
        inequality constraints with only upper bounds:       20

iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
   0  1.0935710e+00 2.80e+00 3.33e-01   0.0 0.00e+00    -  0.00e+00 0.00e+00   0
   1  3.3991294e+00 1.05e+00 1.59e+00  -0.4 2.45e+00    -  3.34e-01 5.92e-01f  1
   2  4.0497209e+00 1.11e-16 3.72e-01  -1.1 7.66e-01    -  1.00e+00 1.00e+00h  1
   3  2.2382925e+00 1.11e-16 1.77e-01  -6.8 5.43e-01    -  9.31e-01 7.65e-01f  1
   4  1.6392751e+00 6.94e-17 1.40e-01  -2.7 4.21e-01    -  1.00e+00 5.15e-01f  1
   5  1.3625675e+00 5.55e-17 9.76e-02  -2.3 2.92e-01    -  9.46e-01 7.53e-01f  1
   6  1.1967681e+00 4.16e-17 9.18e-02  -3.1 2.75e-01    -  9.93e-01 9.60e-01f  1
   7  1.1853197e+00 5.55e-17 6.38e-02  -4.5 1.91e-01    -  1.00e+00 2.51e-01f  1
   8  1.1772459e+00 1.11e-16 7.22e-04  -4.3 2.16e-03    -  1.00e+00 9.99e-01f  1
   9  1.1763524e+00 2.22e-16 4.80e-07 -10.3 1.59e-04    -  9.99e-01 9.96e-01f  1

Number of Iterations....: 9

                                   (scaled)                 (unscaled)
Objective...............:   1.1763523602284711e+00    1.1763523602284711e+00
Dual infeasibility......:   4.7999347130053586e-07    4.7999347130053586e-07
Constraint violation....:   2.2204460492503131e-16    2.2204460492503131e-16
Variable bound violation:   0.0000000000000000e+00    0.0000000000000000e+00
Complementarity.........:   3.7401434460980229e-07    3.7401434460980229e-07
Overall NLP error.......:   4.7999347130053586e-07    4.7999347130053586e-07


Number of objective function evaluations             = 10
Number of objective gradient evaluations             = 10
Number of equality constraint evaluations            = 10
Number of inequality constraint evaluations          = 10
Number of equality constraint Jacobian evaluations   = 10
Number of inequality constraint Jacobian evaluations = 10
Number of Lagrangian Hessian evaluations             = 0
Total seconds in IPOPT                               = 0.942

EXIT: Optimal Solution Found.
Test Summary:       | Pass  Total  Time
solve LP with IPOPT |    1      1  3.1s
Test.DefaultTestSet("solve LP with IPOPT", Any[], 1, false, false, true, 1.742156708288106e9, 1.742156711402232
e9, false, "/home/burger/OCRL/HW3_S25/jl_notebook_cell_df34fa98e69747e1a8f8a730347b8e2f_W5sZmlsZQ==.jl")
```

Part B: Cart Pole Swingup (20 pts)

We are now going to solve for a cartpole swingup. The state for the cartpole is the following:

$$x = [p, \theta, \dot{p}, \dot{\theta}]^T$$

Where p and θ can be seen in the graphic `cartpole.png`.

No description has been provided for this image

where we start with the pole in the down position ($\theta = 0$), and we want to use the horizontal force on the cart to drive the pole to the up position ($\theta = \pi$).

$$
\begin{aligned}
\min_{x_{1:N},u_{1:N-1}} \quad & \sum_{i=1}^{N-1} \left[\frac{1}{2}(x_i - x_{goal})^T Q (x_i - x_{goal}) + \frac{1}{2} u_i^T R u_i \right] + \frac{1}{2}(x_N - x_{goal})^T Q_f (x_N - x_{goal}) && (8)\\
\text{st} \quad & x_1 = x_{\text{IC}} && (9)\\
& x_N = x_{goal} && (10)\\
& f_{hs}(x_i, x_{i+1}, u_i, dt) = 0 \quad \text{for } i = 1, 2, \ldots, N-1 && (11)\\
& -10 \leq u_i \leq 10 \quad \text{for } i = 1, 2, \ldots, N-1 && (12)
\end{aligned}
$$

Where $x_{IC} = [0, 0, 0, 0]$, and $x_{goal} = [0, \pi, 0, 0]$, and $f_{hs}(x_i, x_{i+1}, u_i)$ is the implicit integrator residual for Hermite Simpson (see HW1Q1 to refresh on this). Note that while Zac used a first order hold (FOH) on the controls in class (meaning we linearly interpolate controls between time steps), we are using a zero-order hold (ZOH) in this assignment. This means that each control u_i is held constant for the entirety of the timestep.

In [12]:
```julia
# cartpole
function dynamics(params::NamedTuple, x::Vector, u)
    # cartpole ODE, parametrized by params.

    # cartpole physical parameters
    mc, mp, l = params.mc, params.mp, params.l
    g = 9.81

    q = x[1:2]
    qd = x[3:4]

    s = sin(q[2])
    c = cos(q[2])

    H = [mc+mp mp*l*c; mp*l*c mp*l^2]
    C = [0 -mp*qd[2]*l*s; 0 0]
    G = [0, mp*g*l*s]
    B = [1, 0]

    qdd = -H\(C*qd + G - B*u[1])
    xdot = [qd;qdd]
    return xdot

end
function hermite_simpson(params::NamedTuple, x1::Vector, x2::Vector, u, dt::Real)::Vector
    # TODO: input hermite simpson implicit integrator residual
    xₖ₊ₕ = 0.5 * (x1 + x2) +
            0.125 * dt * (dynamics(params,x1,u) - dynamics(params,x2,u))
    residual_hs = x1 + (dt/6) * (dynamics(params, x1,u) +
        4 * dynamics(params, xₖ₊ₕ,u) + dynamics(params, x2,u)) - x2
    return residual_hs

end
```

```
hermite_simpson (generic function with 1 method)
```

To solve this problem with IPOPT and `fmincon`, we are going to concatenate all of our x's and u's into one vector:

$$Z = \begin{bmatrix} x_1 \\ u_1 \\ x_2 \\ u_2 \\ \vdots \\ x_{N-1} \\ u_{N-1} \\ x_N \end{bmatrix} \in \mathbb{R}^{N\cdot nx+(N-1)\cdot nu}$$

where $x \in \mathbb{R}^{nx}$ and $u \in \mathbb{R}^{nu}$. Below we will provide useful indexing guide in `create_idx` to help you deal with Z.

It is also worth noting that while there are inequality constraints present ($-10 \leq u_i \leq 10$), we do not need a specific `inequality_constraints` function as an input to `fmincon` since these are just bounds on the primal (Z) variable. You should use primal bounds in `fmincon` to capture these constraints.

In [39]:
```
function create_idx(nx,nu,N)
    # This function creates some useful indexing tools for Z
    # x_i = Z[idx.x[i]]
    # u_i = Z[idx.u[i]]

    # Feel free to use/not use anything here.


    # our Z vector is [x0, u0, x1, u1, …, xN]
    nz = (N-1) * nu + N * nx # length of Z
    x = [(i - 1) * (nx + nu) .+ (1 : nx) for i = 1:N]
    u = [(i - 1) * (nx + nu) .+ ((nx + 1):(nx + nu)) for i = 1:(N - 1)]

    # constraint indexing for the (N-1) dynamics constraints when stacked up
    c = [(i - 1) * (nx) .+ (1 : nx) for i = 1:(N - 1)]
    nc = (N - 1) * nx # (N-1)*nx

    return (nx=nx,nu=nu,N=N,nz=nz,nc=nc,x= x,u = u,c = c)
end

function cartpole_cost(params::NamedTuple, Z::Vector)::Real
    idx, N, xg = params.idx, params.N, params.xg
    Q, R, Qf = params.Q, params.R, params.Qf

    # TODO: input cartpole LQR cost

    J = 0
    for i = 1:(N-1)
        xi = Z[idx.x[i]]
        ui = Z[idx.u[i]]

        J += 0.5 * (xi - xg)' * Q * (xi - xg) + 0.5 * ui' * R * ui

    end

    # dont forget terminal cost
    xN = Z[idx.x[N]]
    J += 0.5 * (xN - xg)' * Qf * (xN - xg)

    return J
end

function cartpole_dynamics_constraints(params::NamedTuple, Z::Vector)::Vector
    idx, N, dt = params.idx, params.N, params.dt

    # TODO: create dynamics constraints using hermite simpson

    # create c in a ForwardDiff friendly way (check HW0)
    c = zeros(eltype(Z), idx.nc)

    for i = 1:(N-1)
        xi = Z[idx.x[i]]
        ui = Z[idx.u[i]]
        xip1 = Z[idx.x[i+1]]
```

```
        # TODO: hermite simpson
        c[idx.c[i]] = hermite_simpson(params, xi, xip1, ui, dt)

    end
    return c
end

function cartpole_equality_constraint(params::NamedTuple, Z::Vector)::Vector
    N, idx, xic, xg = params.N, params.idx, params.xic, params.xg

    # TODO: return all of the equality constraints
    x = [Z[idx.x[i]] for i = 1:N]
    return [x[1] - xic; x[end] - xg; cartpole_dynamics_constraints(params, Z)]

    # return x
end

function solve_cartpole_swingup(;verbose=true)

    # problem size
    nx = 4
    nu = 1
    dt = 0.05
    tf = 2.0
    t_vec = 0:dt:tf
    N = length(t_vec)

    # LQR cost
    Q = diagm(ones(nx))
    R = 0.1*diagm(ones(nu))
    Qf = 10*diagm(ones(nx))

    # indexing
    idx = create_idx(nx,nu,N)

    # initial and goal states
    xic = [0, 0, 0, 0]
    xg = [0, pi, 0, 0]

    # load all useful things into params
    params = (Q = Q, R = R, Qf = Qf, xic = xic, xg = xg, dt = dt, N = N, idx = idx,mc = 1.0, mp = 0.2, l = 0.5)

    # TODO: primal bounds
    x_l = -Inf * ones(idx.nz)
    x_u =  Inf * ones(idx.nz)
    for i = 1:N-1
        x_l[idx.u[i]] .= -10
        x_u[idx.u[i]] .= 10
    end


    # inequality constraint bounds (this is what we do when we have no inequality constraints)
    c_l = zeros(0)
    c_u = zeros(0)
    function inequality_constraint(params, Z)
        return zeros(eltype(Z), 0)
    end

    # initial guess
    z0 = 0.001*randn(idx.nz)

    # choose diff type (try :auto, then use :finite if :auto doesn't work)
    # diff_type = :auto
    diff_type = :finite


    Z = fmincon(cartpole_cost,cartpole_equality_constraint,inequality_constraint,
                x_l,x_u,c_l,c_u,z0,params, diff_type;
                tol = 1e-6, c_tol = 1e-6, max_iters = 10_000, verbose = verbose)

    # pull the X and U solutions out of Z
    X = [Z[idx.x[i]] for i = 1:N]
    U = [Z[idx.u[i]] for i = 1:(N-1)]

    return X, U, t_vec, params
```

```
end

@testset "cartpole swingup" begin

    X, U, t_vec = solve_cartpole_swingup(verbose=true)


    # --------------testing------------------
    @test isapprox(X[1],zeros(4), atol = 1e-4)
    @test isapprox(X[end], [0,pi,0,0], atol = 1e-4)
    Xm = hcat(X...)
    Um = hcat(U...)

    # --------------plotting-----------------
    display(plot(t_vec, Xm', label = ["p" "θ" "ṗ" "θ̇"], xlabel = "time (s)", title = "State Trajectory"))
    display(plot(t_vec[1:end-1],Um',label="",xlabel = "time (s)", ylabel = "u",title = "Controls"))

    # meshcat animation
    display(animate_cartpole(X, 0.05))

end
```

```
---------checking dimensions of everything----------
---------all dimensions good-----------------------
---------diff type set to :finite (FiniteDiff.jl)---
---------testing objective gradient----------------
---------testing constraint Jacobian---------------
---------successfully compiled both derivatives-----
---------IPOPT beginning solve---------------------
This is Ipopt version 3.14.17, running with linear solver MUMPS 5.7.3.

Number of nonzeros in equality constraint Jacobian...:    34272
Number of nonzeros in inequality constraint Jacobian.:        0
Number of nonzeros in Lagrangian Hessian.............:        0

Total number of variables............................:      204
                     variables with only lower bounds:        0
                variables with lower and upper bounds:       40
                     variables with only upper bounds:        0
Total number of equality constraints.................:      168
Total number of inequality constraints...............:        0
        inequality constraints with only lower bounds:        0
   inequality constraints with lower and upper bounds:        0
        inequality constraints with only upper bounds:        0

iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
   0  2.4674807e+02 3.14e+00 4.11e-04   0.0 0.00e+00    -  0.00e+00 0.00e+00   0
   1  2.7501452e+02 2.38e+00 7.98e+00  -5.0 1.28e+01    -  4.90e-01 2.43e-01h  3
   2  2.9806244e+02 2.16e+00 1.03e+01  -0.5 1.05e+01    -  6.13e-01 9.26e-02h  4
   3  3.3424994e+02 1.87e+00 1.40e+01  -0.4 1.29e+01    -  6.48e-01 1.33e-01h  3
   4  3.7121036e+02 1.61e+00 2.08e+01  -0.5 1.19e+01    -  8.80e-01 1.40e-01h  3
   5  4.1967396e+02 1.33e+00 2.73e+01  -0.8 1.00e+01    -  1.00e+00 1.74e-01h  3
   6  4.4383472e+02 1.20e+00 3.19e+01   0.3 1.84e+01    -  6.34e-01 9.61e-02h  3
   7  4.7567829e+02 1.07e+00 3.53e+01   0.2 1.80e+01    -  6.49e-01 1.12e-01h  3
   8  5.1183611e+02 9.43e-01 3.90e+01   0.3 2.25e+01    -  6.11e-01 1.17e-01h  3
   9  5.2146844e+02 8.54e-01 3.84e+01   0.3 1.15e+01    -  8.74e-01 9.51e-02h  3
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  10  5.1545768e+02 7.71e-01 4.12e+01   0.4 2.61e+01    -  5.18e-01 9.70e-02f  3
  11  5.0932719e+02 7.01e-01 4.40e+01   0.5 2.68e+01    -  5.99e-01 9.06e-02f  3
  12  5.0693849e+02 6.26e-01 4.92e+01   0.4 3.52e+01    -  8.45e-01 1.07e-01f  3
  13  5.0859557e+02 5.59e-01 5.72e+01   0.6 2.68e+01    -  3.33e-01 1.06e-01h  3
  14  5.3542336e+02 3.51e-01 7.09e+01   0.4 1.99e+01    -  1.94e-01 3.72e-01h  1
  15  5.3515915e+02 2.74e-01 7.29e+01   0.2 1.78e+01    -  2.68e-01 2.19e-01h  1
  16  5.4201437e+02 1.90e-01 7.57e+01   0.7 1.63e+01    -  3.87e-01 3.09e-01f  1
  17  5.3770628e+02 1.18e-01 8.54e+01   0.6 1.18e+01    -  7.65e-01 3.75e-01h  1
  18  5.4061302e+02 8.78e-02 7.83e+01   0.6 9.64e+00    -  8.00e-01 5.53e-01h  1
  19  5.2855467e+02 9.22e-02 5.78e+01   0.4 6.88e+00    -  8.17e-01 9.68e-01h  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  20  5.0258973e+02 3.98e-02 2.14e+01   0.1 2.39e+00    -  9.70e-01 1.00e+00f  1
  21  4.8265684e+02 6.13e-02 2.47e+01   0.1 1.02e+01    -  5.26e-01 3.35e-01f  1
  22  4.6961357e+02 4.33e-02 1.77e+01  -0.1 5.14e+00    -  9.22e-01 4.65e-01f  1
  23  4.6323119e+02 2.27e-01 4.30e+01  -0.1 3.14e+01    -  3.04e-01 2.91e-01f  1
  24  4.6215212e+02 1.63e-01 5.90e+01   0.4 1.91e+01    -  1.00e+00 3.22e-01f  1
  25  4.4565921e+02 7.15e-03 2.80e+01  -0.1 2.80e+00    -  9.93e-01 1.00e+00f  1
  26  4.4070433e+02 5.68e-03 2.00e+01  -0.9 1.74e+00    -  1.00e+00 1.00e+00f  1
  27  4.3812059e+02 2.43e-03 1.89e+01  -1.5 1.36e+00    -  1.00e+00 1.00e+00f  1
  28  4.3720928e+02 2.95e-04 1.78e+01  -2.2 7.29e-01    -  1.00e+00 1.00e+00f  1
  29  4.3662456e+02 3.78e-03 1.86e+01  -2.8 1.47e+00    -  1.00e+00 1.00e+00f  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  30  4.3684720e+02 4.73e-02 1.69e+01  -0.8 1.18e+01    -  2.13e-01 7.19e-01f  1
  31  4.3235942e+02 4.28e-02 1.37e+01  -1.4 4.49e+00    -  1.00e+00 1.00e+00f  1
  32  4.3302222e+02 2.37e-02 1.33e+01  -0.7 3.40e+00    -  1.00e+00 1.00e+00f  1
  33  4.3199234e+02 3.64e-02 1.85e+01  -0.4 2.41e+01    -  7.00e-01 1.39e-01f  3
  34  4.3123136e+02 4.37e-02 2.43e+01  -1.2 2.10e+01    -  3.37e-01 1.19e-01f  3
  35  4.3117797e+02 6.16e-02 3.49e+01  -0.8 1.85e+01    -  1.00e+00 2.04e-01f  2
  36  4.2756052e+02 4.09e-02 3.62e+01  -1.1 4.45e+00    -  1.00e+00 8.58e-01f  1
  37  4.2416042e+02 4.79e-03 3.83e+00  -0.4 1.65e+00    -  9.54e-01 1.00e+00f  1
  38  4.2091671e+02 8.89e-04 5.23e+00  -1.0 2.09e+00    -  9.49e-01 1.00e+00f  1
  39  4.1934826e+02 3.41e-03 1.49e+01  -0.7 7.15e+00    -  7.21e-01 1.00e+00F  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  40  4.1891569e+02 7.88e-03 2.24e+01  -0.9 1.01e+01    -  9.90e-01 2.02e-01f  2
  41  4.2708210e+02 1.66e-02 4.18e+01   0.1 3.50e+01    -  7.22e-01 2.15e-01f  2
  42  4.1977912e+02 2.61e-02 3.57e+01  -0.1 4.85e+00    -  1.00e+00 1.00e+00f  1
  43  4.1900092e+02 3.46e-04 2.76e+01  -0.1 4.55e-01    -  1.00e+00 1.00e+00f  1
  44  4.1653485e+02 2.77e-02 2.43e+01  -0.7 3.35e+00    -  1.00e+00 1.00e+00f  1
  45  4.1529156e+02 5.12e-04 1.19e+01  -0.4 1.74e+00    -  1.00e+00 1.00e+00f  1
  46  4.1398548e+02 1.08e-02 1.72e+01  -0.8 1.89e+00    -  1.00e+00 1.00e+00f  1
```

```
  47  4.1218151e+02 1.89e-02 1.96e+01  -1.7 3.13e+00    -  1.00e+00 9.38e-01f  1
  48  4.1662714e+02 9.77e-02 3.90e+01  -0.1 1.65e+02    -  1.00e+00 3.97e-02f  2
  49  4.1287770e+02 1.09e-02 1.31e+01  -0.3 1.18e+00    -  1.00e+00 1.00e+00f  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  50  4.0988946e+02 9.16e-03 1.66e+01  -0.9 2.08e+00    -  9.93e-01 1.00e+00f  1
  51  4.0787001e+02 7.13e-04 1.43e+01  -1.1 6.12e-01    -  9.97e-01 1.00e+00f  1
  52  4.0690268e+02 8.16e-04 1.58e+01  -2.0 1.59e+00    -  1.00e+00 1.00e+00F  1
  53  4.0686638e+02 1.63e-03 1.95e+01  -1.4 6.72e+00    -  1.00e+00 1.26e-01f  3
  54  4.0698427e+02 1.31e-02 2.90e+01  -1.1 3.26e+01    -  1.00e+00 7.75e-02f  2
  55  4.0667124e+02 1.70e-02 3.34e+01  -1.2 4.58e+01    -  1.67e-01 3.76e-02f  4
  56  4.0840873e+02 6.31e-02 3.97e+01  -0.7 1.18e+01    -  1.00e+00 6.12e-01f  1
  57  4.0407859e+02 2.02e-03 2.71e+01  -0.8 1.98e+00    -  1.00e+00 1.00e+00f  1
  58  4.0309501e+02 8.75e-05 1.86e+01  -1.4 5.01e-01    -  9.98e-01 1.00e+00f  1
  59  4.0264579e+02 2.71e-04 1.38e+01  -1.5 8.63e-01    -  1.00e+00 1.00e+00f  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  60  4.0183401e+02 1.93e-03 1.78e+01  -2.2 2.72e+00    -  1.00e+00 1.00e+00F  1
  61  4.0237564e+02 1.96e-02 3.15e+01  -1.1 1.33e+01    -  1.00e+00 3.00e-01F  1
  62  4.0068263e+02 4.65e-03 4.00e+01  -1.3 6.47e+00    -  1.00e+00 1.00e+00f  1
  63  3.9872004e+02 7.13e-03 1.69e+01  -1.2 2.15e+00    -  9.95e-01 1.00e+00f  1
  64  3.9864494e+02 2.89e-04 9.80e+00  -0.9 7.39e-01    -  1.00e+00 1.00e+00f  1
  65  3.9569348e+02 7.80e-03 1.51e+01  -1.4 3.91e+00    -  1.00e+00 1.00e+00F  1
  66  4.0128923e+02 4.47e-02 3.41e+01  -0.0 2.98e+01    -  7.98e-01 2.09e-01f  2
  67  3.9807938e+02 1.42e-03 1.20e+01  -0.3 1.52e+00    -  1.00e+00 1.00e+00f  1
  68  3.9554269e+02 1.14e-03 5.36e+00  -1.1 7.04e-01    -  9.95e-01 1.00e+00f  1
  69  3.9478424e+02 7.69e-04 4.49e+00  -1.5 6.69e-01    -  9.96e-01 1.00e+00f  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  70  3.9409368e+02 1.69e-03 9.99e+00  -2.3 2.02e+00    -  1.00e+00 1.00e+00F  1
  71  3.9440072e+02 1.89e-03 1.01e+01  -1.9 3.06e+00    -  1.00e+00 5.14e-01H  1
  72  3.9415216e+02 3.60e-03 1.53e+01  -2.2 2.22e+00    -  1.00e+00 1.00e+00f  1
  73  3.9380900e+02 3.71e-03 1.00e+01  -2.3 1.88e+00    -  6.72e-01 1.00e+00f  1
  74  3.9348023e+02 7.57e-05 6.69e+00  -2.5 6.59e-01    -  1.00e+00 1.00e+00f  1
  75  3.9347284e+02 1.99e-04 1.35e+00  -3.4 4.06e-01    -  1.00e+00 1.00e+00f  1
  76  3.9345103e+02 2.54e-05 9.53e-02  -4.3 1.34e-01    -  1.00e+00 9.54e-01h  1
  77  3.9344837e+02 1.66e-07 2.78e-02  -5.8 2.09e-02    -  1.00e+00 9.92e-01h  1
  78  3.9344835e+02 2.02e-07 1.39e-02  -7.2 1.02e-02    -  1.00e+00 1.00e+00h  1
  79  3.9344834e+02 2.70e-09 5.27e-05  -9.1 1.71e-03    -  1.00e+00 1.00e+00h  1
iter    objective    inf_pr   inf_du lg(mu)  ||d||  lg(rg) alpha_du alpha_pr  ls
  80  3.9344834e+02 1.82e-10 5.08e-05 -11.0 3.90e-04    -  1.00e+00 1.00e+00h  1
  81  3.9344834e+02 1.32e-10 6.55e-06 -11.0 2.67e-04    -  1.00e+00 1.00e+00h  1
  82  3.9344834e+02 1.04e-12 1.84e-06 -11.0 3.46e-05    -  1.00e+00 1.00e+00h  1
  83  3.9344834e+02 2.12e-13 9.28e-07 -11.0 2.27e-05    -  1.00e+00 1.00e+00h  1

Number of Iterations....: 83

                                   (scaled)                 (unscaled)
Objective...............:   3.9344833576227290e+02    3.9344833576227290e+02
Dual infeasibility......:   9.2754227920030610e-07    9.2754227920030610e-07
Constraint violation....:   2.1227464230832993e-13    2.1227464230832993e-13
Variable bound violation:   9.9997231828297117e-08    9.9997231828297117e-08
Complementarity.........:   1.0000657567316100e-11    1.0000657567316100e-11
Overall NLP error.......:   9.2754227920030610e-07    9.2754227920030610e-07


Number of objective function evaluations             = 195
Number of objective gradient evaluations             = 84
Number of equality constraint evaluations            = 195
Number of inequality constraint evaluations          = 0
Number of equality constraint Jacobian evaluations   = 84
Number of inequality constraint Jacobian evaluations = 0
Number of Lagrangian Hessian evaluations             = 0
Total seconds in IPOPT                               = 5.384

EXIT: Optimal Solution Found.
```






```
┌ Info: Listening on: 127.0.0.1:8715, thread id: 1
└ @ HTTP.Servers /home/burger/.julia/packages/HTTP/4AUPl/src/Servers.jl:382
┌ Info: MeshCat server started. You can open the visualizer by visiting the following URL in your browser:
│ http://127.0.0.1:8715
└ @ MeshCat /home/burger/.julia/packages/MeshCat/9QrxD/src/visualizer.jl:43
```

Unable to connect

Firefox can't establish a connection to the server at 127.0.0.1:8715.

- The site could be temporarily unavailable or too busy. Try again in a few moments.
- If you are unable to load any pages, check your computer's network connection.
- If your computer or network is protected by a firewall or proxy, make sure that Firefox is permitted to access the web.

Try Again

```
Test Summary:    | Pass  Total  Time
cartpole swingup |    2      2  6.8s
Test.DefaultTestSet("cartpole swingup", Any[], 2, false, false, true, 1.7421601517589e9, 1.742160158579857e9, f
alse, "/home/burger/OCRL/HW3_S25/jl_notebook_cell_df34fa98e69747e1a8f8a730347b8e2f_X12sZmlsZQ==.jl")
```

Part C: Track DIRCOL Solution (5 pts)

Now, similar to HW2 Q2 Part C, we are taking a solution X and U from DIRCOL, and we are going to track the trajectory with TVLQR to account for model mismatch. While we used hermite-simpson integration for the dynamics constraints in DIRCOL, we are going to use RK4 for this simulation. Remember to clamp your control to be within the control bounds.

In [ ]:
```julia
function rk4(params::NamedTuple, x::Vector,u,dt::Float64)
    # vanilla RK4
    k1 = dt*dynamics(params, x, u)
    k2 = dt*dynamics(params, x + k1/2, u)
    k3 = dt*dynamics(params, x + k2/2, u)
    k4 = dt*dynamics(params, x + k3, u)
    x + (1/6)*(k1 + 2*k2 + 2*k3 + k4)
end

@testset "track cartpole swingup with TVLQR" begin

    X_dircol, U_dircol, t_vec, params_dircol = solve_cartpole_swingup(verbose = false)

    N = length(X_dircol)
    dt = params_dircol.dt
    x0 = X_dircol[1]

    # TODO: use TVLQR to generate K's
    nu = size(U_dircol[1],1) # number of inputs
    nx = size(X_dircol[1],1) # number of states

    # use this for TVLQR tracking cost
    Q = diagm([1,1,.05,.1])
    Qf = 100*Q
    R = 0.01*diagm(ones(1))

    # calculate the TVLQR gains for each time step
    K = zeros(nu,nx,N-1)
    P = zeros(nx,nx,N)
    P[:,:,end] = deepcopy(Qf)

    for i = N-1:-1:1
```

```
        A = Matrix(FD.jacobian(x -> rk4(params_dircol, x, U_dircol[i], dt), X_dircol[i]))
        B = Matrix(FD.jacobian(u -> rk4(params_dircol, X_dircol[i], u, dt), U_dircol[i]))
        K[:,:,i] = inv(R + B'*P[:,:,i+1]*B)*B'*P[:,:,i+1]*A
        P[:,:,i] = Q + A'*P[:,:,i+1]*(A - B*K[:,:,i])
    end

    # simulation
    Xsim = [zeros(4) for i = 1:N]
    Usim = [zeros(1) for i = 1:(N-1)]
    Xsim[1] = 1*x0

    # here are the real parameters (different than the one we used for DIRCOL)
    # this model mismatch is what's going to require the TVLQR controller to track
    # the trajectory successfully.
    params_real = (mc = 1.05, mp = 0.21, l = 0.48)

    # TODO: simulate closed loop system with both feedforward and feedback control
    # feedforward - the U_dircol controls that we solved for using dircol
    # feedback - the TVLQR controls
    for i = 1:(N-1)
        # add controller and simulation step
        Usim[i] = clamp.(-K[:,:,i]*(Xsim[i]-X_dircol[i]) + U_dircol[i], -10, 10)
        Xsim[i+1] = rk4(params_real, Xsim[i], Usim[i], dt)
    end


    # -----------------testing----------------------
    xn = Xsim[N]
    @test norm(xn)>0
    @test 1e-6<norm(xn - X_dircol[end])<.8
    @test abs(abs(rad2deg(xn[2])) - 180) < 5 # within 5 degrees
    @test maximum(norm.(Usim,Inf)) <= (10 + 1e-3)

    # -----------------plotting---------------------
    Xm = hcat(Xsim...)
    Xbarm = hcat(X_dircol...)
    plot(t_vec,Xbarm',ls=:dash, label = "",lc = [:red :green :blue :black])
    display(plot!(t_vec,Xm',title = "Cartpole TVLQR (-- is reference)",
                 xlabel = "time (s)", ylabel = "x",
                 label = ["p" "θ" "ṗ" "θ̇"],lc = [:red :green :blue :black]))

    Um = hcat(Usim...)
    Ubarm = hcat(U_dircol...)
    plot(t_vec[1:end-1],Ubarm',ls=:dash,lc = :blue, label = "")
    display(plot!(t_vec[1:end-1],Um',title = "Cartpole TVLQR (-- is reference)",
                 xlabel = "time (s)", ylabel = "u",lc = :blue, label = ""))

    # ----------------animate-----------------------
    display(animate_cartpole(Xsim, 0.05))

end
```






```
┌ Info: Listening on: 127.0.0.1:8716, thread id: 1
└ @ HTTP.Servers /home/burger/.julia/packages/HTTP/4AUPl/src/Servers.jl:382
┌ Info: MeshCat server started. You can open the visualizer by visiting the following URL in your browser:
│ http://127.0.0.1:8716
└ @ MeshCat /home/burger/.julia/packages/MeshCat/9QrxD/src/visualizer.jl:43
```

Unable to connect

Firefox can’t establish a connection to the server at 127.0.0.1:8716.

- The site could be temporarily unavailable or too busy. Try again in a few moments.
- If you are unable to load any pages, check your computer’s network connection.
- If your computer or network is protected by a firewall or proxy, make sure that Firefox is permitted to access the web.

Try Again

```
Test Summary:                 | Pass  Total   Time
track cartpole swingup with TVLQR |    4      4  14.1s
Test.DefaultTestSet("track cartpole swingup with TVLQR", Any[], 4, false, false, true, 1.742160167501927e9, 1.7
42160181570176e9, false, "/home/burger/OCRL/HW3_S25/jl_notebook_cell_df34fa98e69747e1a8f8a730347b8e2f_X14sZmlsZ
Q==.jl")
```